

SIEMENS

PLC— PLC 之间的MPI 通信----调用系统功能的通信方式
communication between PLCs through MPI network -----call SFC

Getting started

Edition 2007-03

摘要 本文简单介绍 S7-300/400 PLC 之间通过 MPI 网络调用系统功能的通信设置

关键词 PLC、MPI、调用系统函数块

Key Words PLC、MPI、Call SFC

目录

1 调用系统功能通信简介	4
2 双向通信.....	4
3 单向通信.....	7

1 调用系统功能通信简介

调用系统功能（SFC65~69）来实现 MPI 的通信,这种通信方式适合于 S7-300/400/200 之间通信，一些非常老的 S7-300/400 CPU 不含有 SFC65~69，所以不能用这种方式通信，只能用全局数据包的方式来通信，判断一个 CPU 是否含有通信的 SFC，可以在联机的情况下，在线查看所用的程序块，看一看是否包含 SFC65~69。通过调用 SFC 来实现通信又可分为两种方式：双向通信和单项通信。以例子的方式介绍通信过程，例子中使用的硬件为 CPU315-2DP，CPU416-2DP。软件为：STEP7 V5.2 SP1。

2 双向通信

在通信的双方都需要调用通信块，一方调用发送块，另一方就要调用接收块来接收数据。这种通信方式适用 S7-300/400 之间通信，发送块是 SFC65(X_SEND)，接收块是 SFC66(X_RCV)。下面以举例的形式说明通信实现的过程：在 STEP7 中创建两个站 STATION1 CPU 416 MPI 站为 2，STATION2 CPU315-2DP MPI 站号为 4，2 号站发送 2 包数据给 4 号站，4 号站判断后放在相应的数据区中。

在 2 号站 OB35 中调用 SFC65，如果扫描时间太短，发送频率太块，对方没有响应，将加重 CPU 的负荷，在 OB35 中调用发送块，发送任务将间隔 100MS 执行一次，编写发送程序如下

```
CALL "X_SEND"                SFC65
  REQ    :=M1.1
  CONT   :=TRUE
  DEST_ID:=W#16#4
  REQ_ID :=DW#16#1
  SD     :=P#DB1.DBX0.0 BYTE 76
  RET_VAL:=MW2
  BUSY   :=M1.2
```

```
CALL "X_SEND"                SFC65
  REQ    :=M1.3
  CONT   :=TRUE
  DEST_ID:=W#16#4
  REQ_ID :=DW#16#2
  SD     :=P#DB2.DBX0.0 BYTE 76
  RET_VAL:=MW4
  BUSY   :=M1.4
```

```
CALL "X_ABORT"              SFC69
  REQ    :=M1.5
  DEST_ID:=W#16#4
  RET_VAL:=MW6
  BUSY   :=M1.6
```

参数中 REQ 为发送请求为 1 时发送。

CONT 为 1 表示连续占用通信资源，为 0 时，通信完成后释放通信资源。

DEST_ID 表示对方的 MPI 地址。

REQ_ID 表示一包数据的标识符，标识符自己定义，例子中两包数据的标识符分别为“1”，“2”。

SD 表示发送区，以指针的格式，例子中第一包数据为 DB1 中从 DBX0.0 (DBB0) 以后 76 个字节，发送区最大为 76 个字节。

RET_VAL 表示发送的状态

BUSY 为 1 时发送中止。

在这个例子中 M1.1，M1.3 为 1 时，CPU416 将发送标识符为“1”和“2”的两包数据给 4 号站 CPU315-2DP。

用户可能会问一个 CPU 究竟可以和能几个 CPU 通信，这和 CPU 的通信资源有关系，这也决定 SFC 的调用的次数，在选项手册中，常常可以看到“动态连接”的个数，这个数字与 SFC 的调用有关，以上例作说明，M1.1，M1.3 为 1 时，与 4 号站的连接建立起来，反之 4

号站发送，2号站接收同样建立一个连接，也就是说两个站通信，都需要发送和接收的情况下占用两个动态连接。

参考图 1:

General	Diagnostic Buffer	Memory	Scan Cycle Time
Time System	Performance Data	Communication	Stacks
Transmission Rate			
Interface X1:	187.5 Kbps		
Connection Resources			
Maximum Number:	12	Not Assigned:	10
	Reserved	Assigned	
PG Communication:	1	1	
OP Communication:	1	0	
S7 Standard Communication:	8	1	
Other Communication:	--	0	

图 1 连接建立

M1.1, M1.3 为 0 时，建立的连接并没有释放，必须调用 SFC69 释放连接，在上例中 M1.5 为 1 时，与 4 号站建立的连接断开，如图 2 所示：

	Reserved	Assigned
PG Communication:	1	1
OP Communication:	1	0
S7 Standard Communication:	8	0
Other Communication:	--	0

图 2 连接释放

有的用户编写多个连接时，由于 CPU 的资源有限，而不能通信，可以用这种方法检测。

在 4 号站编写接收程序如下：

```

CALL "X_RCV"                                SFC66
  EN_DT :=M1.1
  RET_VAL:=MW2
  REQ_ID :=MD4
  NDA   :=M1.2
  RD    :=P#DB1.DBX0.0 BYTE 76

L      MD      4
L      DW#16#1
==D
=      M        1.3

L      MD      4
L      DW#16#2
==D
=      M        1.4

A      M        1.3
JCN    M1
CALL "BLKMOV"                                SFC20
  SRCBLK :=P#DB1.DBX0.0 BYTE 76
  RET_VAL:=MW10
  DSTBLK :=P#DB2.DBX0.0 BYTE 76
M1:   A      M        1.4
      JCN    M2

CALL "BLKMOV"                                SFC20
  SRCBLK :=P#DB1.DBX0.0 BYTE 76
  RET_VAL:=MW12
  DSTBLK :=P#DB3.DBX0.0 BYTE 76

M2:   NOP    0

```

在 OB1 中调用 SFC66 (X_RCV)，参数 EN_DT 表示接收使能，RET_VAL 表示接收状态字，REQ_ID 接收数据包的标识符，NDA 为 1 时指示有新的数据包，为 0 则没有，RD 表示接收区，接收区放在 DB1 中从 DBB0 以后 76 个字节中。例子中，接收块只识别数据的标识符，而不管是哪一个 CPU 发送的，接收从 2 号站 CPU416 发送的两包数据，当标识符为“1”时，M1.3 为 1，复制接收区的数据到 DB2 前 76 个字节中（调用 SFC20），当标识符为“2”时，M1.4 为 1，复制接收区的数据到 DB3 前 76 个字节中。

3 单向通信

与双向通信两方都需要编写发送和接收块不同，单向通信只在一方编写通信程序，这也是客户机与服务器的关系，编写程序一方的 CPU 作为客户机，没有编写程序一方的 CPU 作为服务器，客户机调用 SFC 通信块对服务器的数据进行读写操作，这种通信方式适合 S7-300/400/200 之间通信，S7-300/400 的 CPU 可以同时作为客户机和服务器，S7-200 只能作

服务器。SFC67 (X_GET) 用来读回服务器指定数据区的数据并存放本地的数据区中，SFC68 (X_PUT) 用来写本地数据区的数据到服务器中指定的数据区中，以举例的方式说明怎样调用 SFC 通信块实现通信，先介绍 S7-300/400 通信，与上例一样，建立两个站，STATION1 CPU416-2 MPI 地址为 2 作为客户机，STATION2 CPU315-2DP MPI 地址为 4 作为服务器，调用 SFC68，CPU416 把本地数据区的数据 DB1.DBB0 以后 76 个字节存储在 CPU315 DB1.DBB0 以后 76 个字节中，调用 SFC67，CPU416 读出 CPU315 的数据 DB1.DBB0 以后 76 个字节，放到本地 DB2.DBB0 以后 76 个字节中，例子程序如下：

```

CALL "X_PUT"                SFC68
  REQ      :=M1.1
  CONT     :=TRUE
  DEST_ID  :=W#16#4
  VAR_ADDR:=P#DB1.DBX0.0 BYTE 76
  SD       :=P#DB1.DBX0.0 BYTE 76
  RET_VAL  :=MW2
  BUSY     :=M1.2

CALL "X_GET"                SFC67
  REQ      :=M1.3
  CONT     :=TRUE
  DEST_ID  :=W#16#4
  VAR_ADDR:=P#DB1.DBX0.0 BYTE 76
  RET_VAL  :=MW4
  BUSY     :=M1.4
  RD       :=P#DB2.DBX0.0 BYTE 76

CALL "X_ABORT"             SFC69
  REQ      :=M1.5
  DEST_ID  :=W#16#4
  RET_VAL  :=MW6
  BUSY     :=M1.6

```

参数 DEST_ID 表示对方 MPI 地址，VAR_ADDR 指定服务器的数据区，SD 本地数据区（数据源）M1.1 为 1 时，CPU416 将数据区的数据 DB1.DBB0 以后的 76 个字节存放到 CPU315 的数据区 DB1.DBB0 以后的 76 个字节中。

RD 本地数据接收区，M1.3 为 1 时 CPU416 将 CPU315 数据 DB1.DBB0 以后 76 个字节的数据存放到本地数据区 DB2.DBB0 以后 76 个字节中。数据区最大为 76 个字节。同时在一个 CPU 中调用 SFC67,68 占用一个动态连接，M1.5 为 1 时中断通信释放连接。同样 S7-300CPU 也可以作为客户机，S7-400CPU 也可以作为服务器。

S7-300/400 与 S7-200 通信时在 S7-200 中不能调用 SFC 通信块，只能在 S7-300/400 中调用，所以只有 S7-300/400 可以作为客户机，S7-200 只能作为服务器。下面将以举例的方

式介绍实现通信的过程，例子使用的硬件为 S7-300 CPU315-2DP，S7-200 CPU224，通信卡 CP5611；使用的软件为 STEP7 V5.2 SP1，MICROWIN V3.2。

首先要设定站号和通信速率，S7-300 为 2 号站，S7-200 为 4 号站，通信速率为 187.5K/S，在 S7-300 侧：打开 STEP7，插入一个 S7-300 站，组态站号及通信速率并下载到 CPU 中，在 S7-200 侧：打开 MICROWIN，在“SYSTEM BLOCK”中设定 S7-200 的站号和通信速率，如图 3 所示：




图 3 设置 S7-200 MPI 接口

然后把组态数据下载到 S7-200CPU 中，这样所有的硬件组态作完了，接下来的工作是在 S7-300CPU 中编写通信程序，在 OB1 中调用 SFC68 和 SFC67。程序如下：

```

CALL "X_PUT"                SFC68
  REQ      :=M1.1
  CONT     :=M1.2
  DEST_ID  :=W#16#4
  VAR_ADDR:=QB0
  SD       :=IB0
  RET_VAL  :=MW2
  BUSY     :=M1.3

CALL "X_GET"                SFC67
  REQ      :=M1.4
  CONT     :=M1.5
  DEST_ID  :=W#16#4
  VAR_ADDR:=IB0
  RET_VAL  :=MW4
  BUSY     :=M1.6
  RD       :=QB0

```

M1.1 为 1 时把 S7-300 的输入信号 IB0 发送到 S7-200 的 QB0 中，同时 M1.4 为 1 时把 S7-200 的输入信号 IB0 方到 S7-300 的输出信号 QB0 中。

如果需要与 S7-200 的 V 区通信，在 S7-300CPU 中相对应的是 DB1，例如读出 S7-200 中 VB100 以后 16 个字节并且放到 S7-300 MB20 以后的 16 个字节中，需要调用系统功能 SFC67(X_GET)，在参数 VAR_ADDR 写入 S7-200 的地址区如：P#DB1.DBX100.0 BYTE 16。在参数 RD：写入 S7-300 的地址区如：P#MB20.0 BYTE 16，这样就可以完成通信了。另外还可以通过 S7-200 的 PROFIBUS-DP 模块 EM277 与 S7-300/400 的 MPI 口通信，设置更为简单，在 S7-300/400 侧调用 SFC67/68，与上例参数相同，地址变成 S7-200 EM277 的地址就可以了，在 S7-200 侧，用拨码开关设定 EM277 的站号而不用软件下载设定，连接好以后，从新上电通信速率就可以自适应。

附录一 推荐网址

AS

西门子（中国）有限公司

自动化与驱动集团 客户服务与支持中心

网站首页: <http://www.ad.siemens.com.cn/Service/>

专家推荐精品文档: <http://www.ad.siemens.com.cn/Service/recommend.asp>

AS 常问问题: <http://support.automation.siemens.com/CN/view/zh/10805055/133000>

AS 更新信息: <http://support.automation.siemens.com/CN/view/zh/10805055/133400>

“找答案” AS 版区:

<http://www.ad.siemens.com.cn/service/answer/category.asp?cid=1027>

NET

西门子（中国）有限公司

自动化与驱动集团 客户服务与支持中心

网站首页: <http://www.ad.siemens.com.cn/Service/>

专家推荐精品文档: <http://www.ad.siemens.com.cn/Service/recommend.asp>

Net 常问问题: <http://support.automation.siemens.com/CN/view/zh/10805868/133000>

Net 更新信息: <http://support.automation.siemens.com/CN/view/zh/10805868/133400>

“找答案” Net 版区:

<http://www.ad.siemens.com.cn/service/answer/category.asp?cid=1031>